A Program in BASIC for Calculating Solar Radiation in Tropical Climates on Small Computers

R.H.B. Exell

Division of Energy Technology, Asian Institute of Technology
P.O. Box 2754, Bangkok 10501, Thailand

ABSTRACT

The program consists essentially of a collection of subroutines that may be used for positional calculations and the simulation of daily totals of solar radiation and hourly solar radiation fluxes. Although based on data for Thailand, the model used is expected to be applicable anywhere with a monsoon climate in southeast Asia between latitudes 25°N and 25°S. Provision is made for the separation of the solar radiation fluxes into their direct and diffuse components and for the generation of random sequences of radiation values with the appropriate statistics. The construction of the program is such that it can be used for a wide variety of different purposes.

INTRODUCTION

This paper describes a mathematical model that is being used in the Asian Institute of Technology (AIT) to simulate solar radiation in Thailand for theoretical studies of the performance of solar energy equipment. It is expected to be valid anywhere in South and Southeast Asia with a monsoon climate similar to that of Thailand.

Except for some small modifications, this model is the same as the one presented at the Third Asian School on Solar Energy Harnessing held in AIT during December 1984. It is based on another more complicated model which was developed earlier for use on large computers[1,2] but was later simplified so that it could be used also on small computers without a significant loss of accuracy[3].

The purpose of the present communication is to give in full the program for running the model, and explain how it is used. No attempt is made to discuss the method by which the model was constructed; readers who wish to study the scientific basis of the model should consult references 1, 2 and 3.

GENERAL DESCRIPTION OF THE MODEL

The model provides, among other things, a method for calculating daily totals of global solar radiation and hourly totals of global, direct and diffuse solar radiation under specified conditions of cloudiness. It also enables the user to simulate these quantities as a simple random process with specified degrees of average cloudiness.

A program using the model consists of the following components:

(a) A set of instructions, in lines numbered 0 to 5, for initialising the program (see Listing 1). These instructions must appear at the beginning of every program that uses the model.

Listing 1 Instructions for initializing a program.

```
0 REM SOLAR RADIATION MODEL INITIALIZATION
1 DEG :DIM C(7),PK(9),PKD(9),ZH(12),PSH(12),WH(12),KH(12),GH(12),IH(12),DH(12)
2 C(0)=9:C(1)=36:C(2)=84:C(3)=126:C(4)=126:C(5)=84:C(6)=36:C(7)=9
3 WH(0)=0:WH(1)=0.034:WH(2)=0.066:WH(3)=0.093
4 WH(4)=0.114:WH(5)=0.127:WH(6)=0.132
5 WH(7)=0.127:WH(8)=0.114:WH(9)=0.093:WH(10)=0.066:WH(11)=0.034:WH(12)=0
```

Note: The statement DEG in line 1 instructs the computer to take arguments of trigonometric functions in degrees instead of radians; not every computer has this facility.

(b) A set of main variables through which the user supplies data to the program and obtains results from it. The values of any of these main variables may be set by the user, or the main program may call the subroutines of the model to evaluate them from other main variables that have already been determined earlier in the calculation. These main variables are defined in the section below.

(c) A collection of subroutines, with instructions in lines numbered from 1000 to 1905 and from 10100 to 10113, which calculate the results required by the user. These subroutines are listed and described in detail one by one in "The Subroutines" section.

(d) A set of auxiliary variables, which are used within the subroutines and which should be avoided in the main program.

(e) The main program, which must be written by the user himself incorporating the main variables and subroutines as required. The auxiliary variables and the line numbers in the ranges mentioned above for the initializing instructions and the subroutines must be avoided in the main program. In practice it is convenient to number the lines of short programs in the range 10-990, and the lines of long programs in the range 2000-9990.

THE MAIN VARIABLES

Day of the year: ND. The year is assumed to have exactly 365 days with ND = 1 on 1 January, ND = 2 on 2 January, etc., and ND = 365 on 31 December.

Solar declination (δ): DEL (degrees).

Latitude (ϕ): PHI (degrees). The latitude of the place for which the solar radiation is to be calculated.

Equation of time: E (minutes).

Longitude (λ): LM (degrees East). The longitude of the place for which the solar radiation is to be calculated.

Standard longitude (λ_s): LS (degrees East). Standard longitudes are multiples of 15 degrees east or west of Greenwich.

Zone mean time: ZMT (Hours). The mean time at the standard longitude.

Solar hour angle (ω): OM (degrees).

Solar zenith angle (ζ): ZE (degrees).

Solar azimuth (Ψ): PSI (degrees).

Zenith angle of the normal to a surface (β): BE (degrees). The surface is one that is receiving solar radiation.

Azimuth of the normal to a surface (α): AL (degrees).

Cosine of the angle of incidence ($\cos\theta$): CSTH. The angle θ is the angle of incidence of the solar beam on the surface.

Global solar radiation: G (kW/m^2). Global solar radiation on a horizontal surface.

Direct solar radiation: I (kW/m^2). Direct solar radiation on a surface perpendicular to the solar beam.

Diffuse solar radiation: D (kW/m^2). The diffuse solar radiation on a horizontal surface.

Mean daily clearness: MKD. The mean daily solar radiation divided by the daily solar radiation under a clear sky.

C0, C1, C2, C3, C4. Parameters for calculating MKD by means of the truncated Fourier series.

$$MKD = C0 + C1 \cos(t - C2) + C3 \cos(2t - C4),$$

where t = (ND − 80) 360/365 degrees. A method of calculating these parameters from monthly solar radiation data is given in the Appendix.

Cumulative distribution of daily clearness values: PKD. This is an array of ten probabilities PKD(0), . . . , PKD(9) corresponding to the clearness values 0.05, . . . , 0.95. For example, PKD(3) is the probability that the daily clearness on a particular day is less than or equal to 0.35.

Generalized clearness parameters: K, MK, PK. These are clearness, mean clearness, and cumulative distribution of clearness values, respectively.

Daily total global solar radiation from a clear sky: HC (MJ/m^2).

Hourly clearness: KH. This is an array of thirteen clearness values KH(0), . . ., KH(12), one for each hour of the day from 6:00 to 18:00. For example, KH(7) means the clearness for 13:00.

Hourly position of the sun: ZH, PSH (degrees). These are arrays of thirteen values representing the zenith angle ζ and the azimuth ψ of the sun each hour of the day from 6:00 to 18:00.

Hourly solar radiation: GH, IH, DH (kW/m^2). These are arrays of thirteen values representing the global, normal direct, and diffuse solar radiation each hour of the day from 6:00 to 18:00.

THE SUBROUTINES

The subroutines listed in this section may be called by the user to perform the calculations described. Associated with each subroutine there are variables of several types, as follows:

(a) Input variables, which have to be given values by the user before the subroutine is called.

(b) Auxiliary variables, which are used within the subroutine and should be avoided by the user in the main program.

(c) Output variables, the values of which are calculated by the subroutine.

FIND DELTA (Listing 2)

First line number: 1000.
Input variable: ND.
Auxiliary variables: none.
Output variable: DEL.

This calculates the declination of the sun for a specified day of the year.

Listing 2 Subroutine to find solar declination.

```
1000 REM FIND DELTA
1001 DEL=(ND-80)*0.9863
1002 DEL=0.386+23.273*COS(DEL-92)+0.4*COS(DEL+DEL-19.2):RETURN
```

FIND HOUR ANGLE (Listing 3)

First line number: 1050.
Input variables: LM, LS, ND, ZMT.
Auxiliary variables: none.
Output variables: EQ, OM.

This calculates the hour angle of the sun for a specified longitude, standard longitude, day of the year, and zone mean time. The equation of time EQ is also found.

```
1050 REM FIND HOUR ANGLE
1051 EQ=(ND-80)*0.9863
1052 EQ=0.013+7.342*COS(EQ-194.9)+9.939*COS(EQ+EQ-93.1)
1053 OM=(ZMT-12)*15+LM-LS+EQ*0.25:RETURN
```

FIND SOLAR POSITION (Listing 4)

First line number: 1100.
Input variables: PHI, DEL, OM.

Auxiliary variables: CS, ARG.
Output variables: ZE, PSI.

This calculates the zenith angle and azimuth of the sun for a specified latitude, solar declination, and solar hour angle.

Listing 4 Subroutine to find solar position.

```
1100 REM FIND SOLAR POSITION
1101 ZE=COS(PHI)*COS(DEL)*COS(OM)+SIN(PHI)*SIN(DEL)
1102 CS=ZE:GOSUB 10104:REM ARCCOS
1103 ZE=ARG
1104 IF ZE=0 OR ZE=180 THEN GOTO 1109
1105 SN=COS(DEL)*SIN(OM)/SIN(ZE)
1106 CS=(SIN(PHI)*COS(DEL)*COS(OM)-COS(PHI)*SIN(DEL))/SIN(ZE)
1107 GOSUB 10108:REM ARCTAN2
1108 PSI=ARG:RETURN
1109 PSI=0:RETURN
```

FIND COS THETA (Listing 5)

First line number: 1200
Input variables: AL, BE, ZE, PSI.
Auxiliary variables: CS, SN, ARG.
Output variable: CSTH.

This calculates the cosine of the angle of incidence of the direct solar beam on a surface for a specified orientation of the surface and position of the sun.

Listing 5 Subroutine to find cosine of angle of incidence of solar beam on a surface.

```
1200 REM FIND COS THETA
1201 CSTH=COS(ZE)*COS(BE)+SIN(ZE)*SIN(BE)*COS(PSI-AL):RETURN
```

FIND SOLAR RADIATION (Listing 6)

First line number: 1300.
Input variables: ND, ZE, K.

Listing 6 Subroutine to find solar radiation fluxes.

```
1300 REM FIND SOLAR RADIATION
1301 IF ZE>=90 THEN GOTO 1307
1302 Z=ZE*ZE/8100:F=1-0.0335*SIN(0.9863*(ND-94))
1303 G=((((((-4.4631*Z+13.0798)*Z-13.899)*Z+6.6849)*Z-1.072)*Z-1.4354)*Z+1.1049)
*F*K
1304 I=((((((-0.9217*Z+3.7206)*Z-5.7674)*Z+3.3705)*Z-1.1883)*Z-0.2001)*Z+0.9864)
*F*K
1305 R=1:S=K*K:IF ZE<90 THEN R=1-(I/G)*COS(ZE):D=(1-(S*(0.733+S*0.267))*(1-R))*G
1306 I=0:IF ZE<=87.5 THEN I=(G-D)/COS(ZE):RETURN
1307 G=0:I=0:D=0:RETURN
```

Auxiliary variables: F, R, S, Z.
Output variables: G, I, D.

This calculates the global, normal direct, and diffuse solar radiation for a specified day of the year, solar zenith angle and clearness.

FIND MEAN KD AND PKD (Listing 7)

First line number: 1400.
Input variables: CO, C1, C2, C3, C4, ND.
Auxiliary variables: MK, P, PK, S.
Output variables: MKD, PKD(0), . . . , PKD(9).

This calculates the mean clearness for a specified day of the year, and the corresponding cumulative distribution of clearness values. The Fourier parameters for the mean daily clearness must also be specified.

Listing 7 Subroutine to find mean daily clearness from specified data.

```
1400 REM FIND MEAN KD AND PKD
1401 MKD=(ND-80)*0.9863
1402 MKD=CO+C1*COS(MKD-C2)+C3*COS(MKD+MKD-C4)
1403 MK=MKD:GOSUB 1500:REM FIND PK
1404 FOR S=0 TO 9:PKD(S)=PK(S):NEXT S
1405 RETURN
```

FIND PK (Listing 8)

First line number: 1500.
Input variable: MK.
Auxiliary variables: C(0), . . . , C(7), P, S.
Output variables: PK(0), . . ., PK(9).

This calculates a cumulative probability distribution of clearness values for a specified general mean clearness.

Listing 8 Subroutine to find probability distribution of clearness values.

```
1500 REM FIND PK
1501 IF MK<0.05 THEN MK=0.05
1502 IF MK>0.95 THEN MK=0.95
1503 P=(MK-0.05)/0.9:PK(0)=(1-P)^9:PK(9)=P^9
1504 FOR S=1 TO 8:PK(S)=C(S-1)*P^S*(1-P)^(9-S):NEXT S
1505 FOR S=1 TO 9:PK(S)=PK(S-1)+PK(S):NEXT S
1506 RETURN
```

FIND K (Listing 9)

First line number: 1600.
Input variables: PK(0), . . . , PK(9).

Auxiliary variables: R, JK.
Output variable: K.

This selects a value of K at random in accordance with a specified cumulative probability distribution of values.

Listing 9 Subroutine to pick a clearness value at random.

```
1600 REM FIND K
1601 R=RND(0):K=0.05
1602 FOR JK=0 TO 8:IF R>PK(JK) THEN K=K+0.1
1603 NEXT JK
1604 RETURN
```

Note: The statement R=RND(0) in line 1601 assigns a random number to R such that $0 \leqslant R < 1$; on different computers this has to be written in different ways.

FIND DAILY CLEAR SKY RADIATION (Listing 10)

First line number: 1700.
Input variables: PHI, ND.
Auxiliary variables: F, H1, H2, H3.
Output variable: HC.

This calculates the daily total global solar radiation from a clear sky for a specified latitude and day of the year. It is valid for latitudes between 25°N and 25°S.

Listing 10 Subroutine to find daily solar radiation under a clear sky.

```
1700 REM FIND DAILY CLEAR SKY RADIATION
1701 HC=(ND-80)*0.9863:F=1-0.0335*SIN(0.9863*(ND-94))
1702 H1=(-4.1215E-03*PHI+3.25E-03)*PHI+27.4486:H2=0.321*PHI
1703 H3=(-6.025E-04*PHI+2.4E-03)*PHI+1.215:HC=(H1+H2*COS(HC-92)+H3*COS(HC+HC-0.1
2*PHI-3.4))*F
1704 RETURN
```

FIND KH (Listing 11)

First line number: 1800.
Input variable: KD.

Listing 11 Subroutine to pick hourly clearness values.

```
1800 REM FIND KH
1801 MK=KD:GOSUB 1500:REM FIND PK
1802 S=0
1803 FOR H=0 TO 12:GOSUB 1600:REM FIND K
1804 KH(H)=K:S=S+KH(H)*WH(H):NEXT H
1805 IF ABS(S-KD)>=0.05 THEN GOTO 1802
1806 RETURN
```

Auxiliary variables: MK, H, S, WH(0), . . ., WH(12), all variables in FIND PK, all variables in FIND K.

Output variables: KH(0), . . ., KH(12).

This selects values of the hourly clearness at random so as to give a suitably weighted average close to a specified daily clearness.

FIND HOURLY RADIATION (Listing 12)

First line number: 1900.

Input variables: PHI, DEL, ND, KH(0), . . ., KH(12).

Auxiliary variables: H, all variables in FIND SOLAR POSITION, all variables in FIND SOLAR RADIATION.

Output variables: ZH(H), PSH(H), GH(H), IH(H), DH(H), where H = 0, . . ., 12.

This calculates the solar zenith angle and azimuth, and the global, normal direct, and diffuse solar radiation each hour for specified latitude, day of the year, solar declination, and hourly clearness values.

Listing 12 Subroutine to find hourly solar radiation fluxes.

```
1900 REM FIND HOURLY RADIATION
1901 FOR H=0 TO 12:OM=H*15-90:GOSUB 1100:REM FIND SOLAR POSITION
1902 ZH(H)=ZE:PSH(H)=PSI:K=KH(H):GOSUB 1300:REM FIND SOLAR RADIATION
1903 GH(H)=G:IH(H)=I:DH(H)=D
1904 NEXT H
1905 RETURN
```

ARCSIN (Listing 13)

First line number: 10100.

This calculates the principle value ARG of arcsin(SN) in degrees.

Listing 13 Subroutine to find arcsine of a value.

```
10100 REM ARCSIN
10101 IF ABS(SN)<0.999999 THEN GOTO 10103
10102 ARG=90*SGN(SN):RETURN
10103 ARG=ATN(SN/SQR(1-SN*SN)):RETURN
```

ARCCOS (Listing 14)

First line number: 10104.

This calculates the principle value ARG of arccos (CS) in degrees.

Listing 14 Subroutine to find arccosine of a value.

```
10104 REM ARCCOS
10105 IF ABS(CS)<0.999999 THEN GOTO 10107
10106 ARG=90-90*SGN(CS):RETURN
10107 ARG=90-ATN(CS/SQR(1-CS*CS)):RETURN
```

ARCTAN2 (Listing 15)

First line number: 10108.

This calculates the argument ARG such that SN = sin (ARG) and CS = cos (ARG); the result is in degrees in the correct quadrant.

Listing 15 Subroutine to find the argument giving a specified sine and cosine.

```
10108 REM ARCTAN2
10109 IF ABS(CS)<1E-06 THEN GOTO 10113
10110 ARG=ATN(SN/CS):IF CS<0 THEN ARG=ARG-180*SGN(ARG)
10111 IF CS<0 AND ARG=0 THEN ARG=180
10112 RETURN
10113 ARG=90*SGN(SN):RETURN
```

RELATIONSHIPS BETWEEN VARIABLES

The relationships between variables and subroutines that may be used to calculate some variables from others are shown in Fig. 1. Each main variable may be specified by the user of the model, or it may be calculated by using the appropriate subroutine. The numbers above the variables are the line numbers of the subroutines used to calculate them, and the arrows pointing at the line numbers are from those variables that must be specified before the subroutine is called. Variables that are used together in the calculations are enclosed in boxes.



Fig. 1 Relations between main variables. Numbers indicate first lines of subroutines. A broken arrow is drawn with the relation is stochastic. Note that EQ is not required to find ZE and PSI, and PSI is not required to find G,I and D.

For example, consider the calculation of the solar zenith angle ZE and the solar azimuth PSI by means of the subroutine FIND SOLAR POSITION at line 1100. Before calling this subroutine one must assign values to PHI, DEL and OM. These values may be chosen at will by the user for a particular purpose. Alternatively, DEL may be calculated from the day of the year ND by means of the subroutine FIND DELTA. Likewise, the solar hour angle OM may be assigned at will by the user, or it may be calculated by means of the subroutine FIND HOUR ANGLE at line 1050 from pre-assigned values of the day of the year ND, the longitude LM, the standard longitude LS, and the zone mean time ZMT.

EXAMPLE OF A MAIN PROGRAM

Listing 16 shows a simple main program that generates a fluctuating sequence of values for the daily global solar radiation HD simulating the climate of Bangkok.

Listing 16 Example of a main program using the model.

```
10 C0=0.636:C1=0.099:C2=302:C3=0.023:C4=139:PHI=13.7
20 FOR NW=8 TO 365 STEP 14
30 GOSUB 1400:REM FIND MKD
40 MK=MKD:GOSUB 1500:REM FIND PK(S)
45 FOR ND=NW-7 TO NW+6
50 GOSUB 1600:REM FIND K
60 GOSUB 1700:REM FIND HC
70 HD=HC*K:PRINT ND,HD
100 NEXT ND
105 NEXT NW
110 END
```

To do this the program first sets values of the Fourier coefficients in the formula for MKD, and the latitude 13.7°N of Bangkok (line 10). Next it calculates MKD and the corresponding probability distribution of clearness values every 14 days (lines 20-40). Then for each 14-day period it calculates the daily total global solar radiation under a clear sky, multiplies it by a randomly selected clearness value, and prints the result (lines 45-70). Finally the loops are closed and the program is terminated (lines 100-110).

REFERENCES

1. Exell, R.H.B. (1980). *Simulation of solar radiation in a tropical climate with data for Thailand.* AIT Research Report No. 115.
2. Exell, R.H.B. (1981). A mathematical model for solar radiation in South-east Asia (Thailand). *Solar Energy,* Vol. 26, pp. 161-168.
3. Exell, R.H.B. (1984). A simple statistical model of daily global solar radiation in Thailand. *J. Sci. Soc. Thailand,* Vol. 10, pp. 53-56.

APPENDIX

The parameters C0, C1, C2, C3 and C4 for calculating the mean daily clearness may be found by the following steps:

(1) Calculate the daily solar radiation HC from a clear sky for the middle of each month.

(2) Divide the mean daily solar radiation, as determined from observational data, by HC to find the mean daily clearness each month.

(3) Obtain the parameters by numerical harmonic analysis of the mean daily clearness values.

Listing 17 gives a program for performing these calculations. The monthly means of daily solar radiation in units MJ/m^2 are supplied as input, and the parameters are printed as output.

Listing 17 Program to find Fourier Series parameters from solar radiation data.

```
0 REM CALCULATION OF FOURIER SERIES PARAMETERS
5 DATA 15,46,74,105,135,166,196,227,258,288,319,349
10 PRINT "ENTER LATITUDE":INPUT PHI
20 DIM Y(11),A(2),B(2):DEG
25 PRINT "ENTER MONTHLY MEANS OF DAILY RADIATION"
30 FOR K=0 TO 11:INPUT I
35 Y(K)=I:READ ND:GOSUB 1700
40 Y(K)=Y(K)/HC:NEXT K
50 FOR I=0 TO 2
55 A(I)=0:B(I)=0
60 FOR K=0 TO 11
70 A(I)=A(I)+Y(K)*COS(30*K*I)/6
80 B(I)=B(I)+Y(K)*SIN(30*K*I)/6
90 NEXT K
100 NEXT I
110 CO=A(0)/2
120 C1=SQR(A(1)*A(1)+B(1)*B(1))
130 SN=-B(1)/C1:CS=A(1)/C1
140 GOSUB 10108:REM ARCTAN2
150 C2=-ARG:C2=C2-64.11
160 C3=SQR(A(2)*A(2)+B(2)*B(2))
170 SN=-B(2)/C3:CS=A(2)/C3
180 GOSUB 10108:REM ARCTAN2
190 C4=-ARG:C4=C4-128.22
200 PRINT CO,C1,C2,C3,C4:END
1700 REM FIND DAILY CLEAR SKY RADIATION
1701 HC=(ND-80)*0.9863:F=1-0.0335*SIN(0.9863*(ND-94))
1702 H1=(-4.1215E-03*PHI+3.25E-03)*PHI+27.4486:H2=0.321*PHI
1703 H3=(-6.025E-04*PHI+2.4E-03)*PHI+1.215:HC=(H1+H2*COS(HC-92)+H3*COS(HC+HC-0.1
2*PHI-3.4))*F
1704 RETURN
10108 REM ARCTAN2
10109 IF ABS(CS)<1E-06 THEN GOTO 10113
10110 ARG=ATN(SN/CS):IF CS<0 THEN ARG=ARG-180*SGN(ARG)
10111 IF CS<0 AND ARG=0 THEN ARG=180
10112 RETURN
10113 ARG=90*SGN(SN):RETURN
```